
Esercizio

Si consideri un punto P di massa m mobile lungo una guida rettilinea liscia
verticale e sia x la coordinata di P lungo un asse parallelo alla guida ed orientato
verso il basso. Sul punto, oltre al peso ed alla reazione vincolare, agiscono una
forza di attrito di tipo viscoso (proporzionale alla velocità) ed una forza elastica
non lineare. L’equazione del moto di P si ottiene proiettando lungo la guida
l’equazione di Newton e risulta:

mẍ = mg − α ẋ+ fel(x) (1)

dove g è l’accelerazione di gravità, α il coefficiente (positivo) di attrito viscoso e

fel(x) = −k (x+ c x3) k, c > 0

la forza elastica.

• Dimostrare che il sistema ha una sola configurazione di equilibrio, xeq, e
che xeq può essere approssimato con il metodo di bisezione.

• Posto g = 9.81 m/s2, m = 1 kg, k = 10 N/m, c = 10−2 m−2 e α = 0.7
Ns/m, utilizzare la procedura definita in bisezione.sci1 per determinare
un’approssimazione di xeq con errore assoluto inferiore a 10−5m.

• Posto z = x− xeq, scrivere l’equazione del moto (1) in termini di z.

• Nell’equazione “(differenziale non lineare) scritta al punto precedente, com-
pare l’addendo fel(xeq + z). Sostituire tale termine con il polinomio di
Taylor di ordine uno:

p(z) = fel(xeq) + f ′
el(xeq) z

e scrivere la corrispondente equazione differenziale (lineare ed omogenea).
Le soluzioni di quest’ultima approssimano i moti del punto “per piccoli
scostamenti dal moto di equilibrio x(t) = xeq.”

• Si consideri l’equazione differenziale lineare ottenuta nel punto precedente.
La soluzione dell’equazione che verifica le condizioni iniziali z(0) = 0.1 m e
ż(0) = 0 m/s può essere approssimata, sull’intervallo di tempo [0, 5] s, con
il metodo di Eulero. Utilizzare la procedura definita in LMV TS 1 pv.sci

imponendo un errore locale massimo consentito pari a 10−1. Determinare
il numero di passi N1 effettuati dalla procedura ed utilizzare il comando
plot per ottenere i grafici della posizione (z) e della velocità (ż) in funzione
del tempo.

1I files .sci a cui si fa riferimento si trovano sulla pagina web del corso nella sezione altro
materiale didattico.

• Verificare che l’equazione differenziale lineare descrive il moto di un oscil-
latore armonico smorzato. Ci si aspetta quindi che le funzioni posizione e
velocità siano oscillazioni di ampiezza decrescente nel tempo. Si constata
invece che le soluzioni numeriche hanno andamento opposto (oscillazioni di
ampiezza crescente nel tempo). Giustificare questa differenza qualitativa.

• Ripetere l’integrazione numerica imponendo un errore locale massimo con-
sentito pari a 10−i con i = 2, . . . , 5. Per ciascuna delle integrazioni, de-
terminare il numero di passi Ni che la procedura ha effettuato ed i valori
massimo e minimo del passo utilizzati.

• Calcolare i rapporti Ni+2/Ni per i = 1, 2 e 3. Giustificare i risultati ot-
tenuti ed indicare quale dovrebbe essere, approssimativamente, il numero
di passi che la procedura effettua imponendo un errore locale massimo
consentito pari a 10−9.

//
clear;
//
Percorso = "/home/ciampa/Lavoro/Scilab/Lavoro/LMV/";
//
// costanti
//
g = 9.81;
m = 1;
k = 10;
c = 1d-2;
alfa = 0.7;
//
// STATICA ****************************
//
function y=G(x)
    y = g - (k/m)*(x + c*x.^3)
endfunction
//
exec(Percorso + 'Zeri/bisezione.sci');
//
a = 0;
b = g/10;
tol = 1d-5;
kmax = 60;
//
[z,v,info] = bisezione(G,a,b,"assoluto",tol,kmax,"zitto");
//
printf("z = ");disp(z);
printf("v = ");disp(v);
printf("info = ");disp(info);
//
xeq = z;
B = -k*(1 + 3*c*xeq^2);
A = [  0,       1;
     B/m, -alfa/m];
//
// DINAMICA LINEARE ****************************
//
function y=d1x(t, x)
    y = A * x
endfunction
//
function y=d2x(t, x)
    y = A * d1x(t,x)
endfunction
//
exec(Percorso + 'ODE/LMV_TS_1_pv.sci');
//
x0 = [10;0];
t0 = 0;
tf = 5;
EL_MAX = 1d-4;
//
[T,U,PASSO] = LMV_TS_1_pv(x0,t0,tf,d1x,d2x,EL_MAX,"zitto");
//
rad = spec(A);
si = real(rad(1));
om = abs(imag(rad(1)));
//
function y=xe(t)
    y = [x0(1)*exp(si*t).*(-(si/om)*sin(om*t)+cos(om*t));
        -x0(1)*((si^2+om^2)/om)*exp(si*t).*sin(om*t)]
endfunction
//
et = U - xe(T);
ET(1) = 0;
for i=2:length(T),
    ET(i) = norm(et(:,i));
end;
printf("\nerrore totale massimo = %3.2e\n",max(ET));



//
// grafici
//
scf(0); clf();
//
subplot(221);
plot(T,U(1,:),"r",T,U(2,:),"b");
legend("posizione LIN","velocità LIN");
xgrid();
//
subplot(222);
plot(T(1:$-1),PASSO,"b");
xgrid();
xtitle("passo LIN");
//
//
// DINAMICA NON LINEARE ****************************
//
function y=f_el(x)
    y = -k*(x + c*x^3)
endfunction
//
function y=d1x_nl(t, x)
    y = [x(2);
         -(alfa/m)*x(2) + f_el(xeq + x(1))/m + g]
endfunction
//
function y=d2x_nl(t, x)
    z = -(alfa/m)*x(2) + f_el(xeq + x(1))/m + g
    y = [z;
         -(alfa/m)*z + (-k + 3*c*(xeq + x(1))^2)*x(2)/m]
endfunction
//
EL_MAX = 1d-6;
//
[T_nl,U_nl,PASSO_nl] = LMV_TS_1_pv(x0,t0,tf,d1x_nl,d2x_nl,EL_MAX,"zitto");
//
// grafici
//
subplot(223);
plot(T_nl,U_nl(1,:),"r",T_nl,U_nl(2,:),"b");
legend("posizione NL","velocità NL");
xgrid();
//
subplot(224);
plot(T_nl(1:$-1),PASSO_nl,"b");
xgrid();
xtitle("passo NL");
//


